Limits, Infinity, and Asymptotes

Objective

There are three objectives of this lab assignment: i) to develop your ability to determine limits at ±∞, ii) to recognize when a limit diverges to ±∞, and iii) to use limits at infinity and infinite limits to determine asymptotes for the graph of a function.

Background

There are three types of asymptotes: horizontal, vertical, and oblique.

<table>
<thead>
<tr>
<th>Type</th>
<th>Equation</th>
<th>Defining Property</th>
</tr>
</thead>
<tbody>
<tr>
<td>Horizontal</td>
<td>( y = b )</td>
<td>( \lim_{x \to \infty} f(x) = b ), ( \lim_{x \to -\infty} f(x) = b )</td>
</tr>
<tr>
<td>Vertical</td>
<td>( x = a )</td>
<td>( \lim_{x \to c^+} f(x) = \infty ), ( \lim_{x \to c^-} f(x) = \infty ), ( \lim_{x \to c^+} f(x) = -\infty ), ( \lim_{x \to c^-} f(x) = -\infty )</td>
</tr>
<tr>
<td>Oblique</td>
<td>( y = ax + b )</td>
<td>( \lim_{x \to \infty} (f(x) - (ax + b)) = 0 ), ( \lim_{x \to -\infty} (f(x) - (ax + b)) = 0 )</td>
</tr>
</tbody>
</table>

The Asymptotes command generally returns all asymptotes — horizontal, vertical, or oblique — for a function. This command is available only after loading the Student[Calculus1] package. The Asymptotes command is implemented using Maple’s capabilities to evaluate limits, to determine singularities of functions, and to perform various symbolic manipulations (such as long division of polynomials).

The Asymptotes command returns the asymptotes as a list of equations. In this form the implicitplot command is the easiest way to plot equations (not expressions). Unfortunately, in Maple 8 the implicitplot command can accept only a set of equations. The convert command can be used to change a list into a set. Here it is simpler to construct the set explicitly with \{ expr1, expr2 \}.

The limit command is all that is needed to determine any horizontal asymptotes for a function. Note that the mathematical constant \( \infty \) is called infinity in Maple.

Vertical asymptotes can often be found by determining the zeros of the denominator of a function. The numer and denom commands are used to obtain the numerator and denominator of a rational expression. Then, factor or solve can be used to identify the zeros of an appropriate expression.

The quo and rem commands perform polynomial division that is frequently needed to determine oblique asymptotes.

Discussion

Enter, and execute, the following Maple commands in a Maple worksheet:

Example 1: Asymptotes command

```maple
> restart; # clear Maple’s memory
> with( plots ); # load plots package
> with( Student[Calculus1] ); # load package
> w := x -> (2*x^5+3*x^3-2*x-2)/(x^4-1); # define function
> Pw := plot( w(x), x=-10..10, y=-20..20, discont=true ); # plot function
> y=-20..20, discont=true ): # display plot
> Pw;
> asym := Asymptotes( w(x), x ); # asymptotes as list
> Pa1 := implicitplot( asym[1], x=-10..10, y=-20..20, linestyle=2 ): # plot of oblique asymp
> Pa2 := implicitplot( {asym[2], asym[3]}, x=-10..10, y=-20..20, linestyle=3 ): # plot of vertical asymps
> display( [ Pw, Pa1, Pa2 ] ); # display combined plot
```

Maple Lab for Calculus I Fall 2003
Example 2: Horizontal Asymptotes

```maple
> g := x -> (x^4-2*x^3+2*x-1)/(x^4+1);
# define rational function
> Pg := plot( g(x), x=-20..20 );
# create graph of function
> Pg;
# display graph of function
> q1 := limit( g(x), x=infinity );
# horizontal asymptote?
> q2 := limit( g(x), x=-infinity );
# horizontal asymptote?
> horiz := { q1, q2 };  # set of horizontal asymptotes
> Ph := plot( horiz, x=-20..20, color=cyan );  # create graph of horiz asymp
> display( [ Pg, Ph ] );  # display combined graph
```

Example 3: Vertical Asymptotes

```maple
> f := x -> (sin(x)-cos(x)+1)/(x^3-3*x+2);
# define function
> Pf := plot( f(x), x=-4..4,
    y=-10..10, discont=true );
# create graph of function
> Pf;
# display graph of function
> q1 := denom( f(x) );
# denominator of f(x)
> q2 := solve( q1=0, {x} );
# locate singularities
> vert := { x=-2, x=1 };
# vertical asymptotes
> Pv := implicitplot( vert, x=-2*Pi..2*Pi,
    y=-20..20, color=blue );
# create graph of vert asymp
> display( [ Pf, Pv ] );
# display combined graph
```

Example 4: Oblique Asymptotes (for Rational Functions)

```maple
> u := x -> (3*x^3-4*x^2-5*x+3)/(x^2+1);
# define rational function
> Pu := plot( u(x), x=-10..10 );
# create plot
> Pu;
> q1 := numer( u(x) );
# numerator of function
> q2 := denom( u(x) );
# denominator of function
> q3 := quo( q1, q2, x );
# quotient from long division
> q4 := rem( q1, q2, x );
# remainder from long division
> u2 := q3 + q4/q2;
# equivalent form of u
> u(x) = simplify( u2 );
# equivalent expressions?
> Po := plot( q3, x=-10..10, color=pink );
# create plot of oblique asymp
> display( [ Pu, Po ] );
# display combined plot
```

Notes

(1) In Example 1, the different types of asymptotes are distinguished with different linestyle options. When a plot will be printed in black-and-white, this is preferable to using the color option.

Questions

(1) Use the limit command to explain why there are no horizontal asymptotes in Example 4.

(2) Find all horizontal, vertical, and oblique asymptotes for \( f(x) = \frac{2|x^3+3}{x^3+1} - \frac{8\sin x}{x^2+1} \). List the asymptotes and include a clearly labeled graph of the function and its asymptotes.

(3) (a) Write the function in Example 1 in the form \( w(x) = L(x) + R(x) \) where \( L \) is a linear function and \( R \) is the ratio of two polynomials for which the numerator has a smaller degree than the denominator. Write the denominator of \( R(x) \) in factored form.
(b) Explain how (a) allows the vertical asymptotes of \( w(x) \) to be determined by inspection.
(c) Show that the graph of \( R(x) \) has \( y = 0 \) as its horizontal asymptote.